
Assigned: 06-Oct-2009                        Due: 15-Oct-2009 I m p o r t a n t : P l e a s e t u r n - i n a l l o f y o u r c o d e s a l o n g w i t h t h e w a v e f o r m s ( w h e n n e c e s s a r y )d u r i n g s u b m i s s i o n o f y o u r s o l u t i o n . Y o u m a y b e a s k e d t o p r o v i d e t h e s o f t c o p y o f y o u r c o d e sb y e m a i l i f s u c h n e e d a r i s e s .P r o b l e m 1 :
For a gate if we increase the widths of the transistors, the corresponding 

propagation delays get decreased first but if you continue to increase the widths, you will find 

that the propagation delays get increased eventually.  P a r t a )
Explain why does it happen.  

Perform SPICE simulations for a simple CMOS inverter with the libraries that you have used in 

lab, i.e., t s m c 2 5 N  and t s m c 2 5 P  to confirm your answer. Use L = 300 nm, VDD = 2.5 V.  P a r t b )
A gate has r i s i n g  and f a l l i n g  propagation delays. Also a gate with more than one input 

can have different rising and falling propagation delays depending on different input vector 

combinations applied to the inputs of the gate. Consider a 2-input NAND gate. Explain for which 

input combinations, the m a x i m u m r i s i n g  and the m a x i m u m f a l l i n g  delays occur.  

Perform SPICE simulation with the libraries that you have used in lab, i.e., t s m c 2 5 N  and t s m c 2 5 P  to confirm your answer. Use L = 300 nm, VDD = 2.5 V. P a r t c )
Find out the o p t i m u m w i d t h s o f t h e t r a n s i s t o r s  so that the m a x i m u m i n p u t - t o - o u t p u tp r o p a g a t i o n d e l a y f o r a 2 - i n p u t N A N D g a t e  is m i n i m i z e d . Clearly explain the procedure you 

have followed. Submit the necessary plots showing the trade-off between propagation delays 

and transistor widths. S o l u t i o nP a r t a )
As the width of a transistor in a gate is increased, the corresponding ON current increases. 

Accordingly, the corresponding propagation delay decreases. But the capacitance also increases 

proportionally with the increase of the transistor width at the output node (as well as at the 

internal nodes, if any). Increase of capacitance attributes to the increase of propagation delay 

(for an inverter both the high-to-low and low-to-high propagation delays are increased as the 

output node capacitance increases). However, for example, for an inverter if PMOS transistor 

width is increased, the increase of low-to-high propagation delay due to the increase of output 

node capacitance c a n  get compensated due to the increase of ON current and the trend will 

depend on the libraries/models you are using.  Accordingly, a trade-off is expected to happen 

for some configuration of the transistors widths in a gate beyond which the propagation delay 

decreases with the increase of transistor widths due to the increase of ON current and over 

which the propagation delay increases because of the increase of capacitances. We can use the 

average of the high-to-low and low-to-high propagation delays since it captures both the 

increase of ON current and the increase of capacitance with the increase of transistor widths. 

The cadence technology library that we are using restricts us to keep a 75 nm difference 

between two consecutive widths. So we need to choose the set of values in the .DATA 

statement (in the HSPICE netlist) accordingly. H S P I C E C o d e
* HW 3, Problem 1 Part a) Solution 

* ECE 559, Fall 2009, Purdue University 

 

.GLOBAL VDD! 

 

.lib "$CDK_DIR/models/hspice/public/publicModel/tsmc25N" NMOS  

.lib "$CDK_DIR/models/hspice/public/publicModel/tsmc25P" PMOS 

 

.PARAM VDD  = 2.5v 

.PARAM VREF = 1.25v 

.PARAM L    = 300n 

 

.PARAM WP   = 450n 

.PARAM WN   = 450n 

 

* fanout of 4 is chosen 

.PARAM WPF  = '4*WP' 

.PARAM WNF  = '4*WN' 

 

.TEMP    25.0000 

.OPTION  POST    

 

* set of data for simulation 

.DATA data1 

 WN WP 

 450n 450n 

 450n 525n 

 450n 600n 

 450n 675n 

* put all set of values that you want to simulate 

.ENDDATA 

 

MP1 OUT IN VDD! VDD!  TSMC25P  L='L' W='WP'  

+AD='WP*2.5*L' AS='WP*2.5*L' PD='2*WP+5*L' PS='2*WP+5*L' M=1  

MN1 OUT IN 0 0  TSMC25N  L='L' W='WN'  

+AD='WN*2.5*L' AS='WN*2.5*L' PD='2*WN+5*L' PS='2*WN+5*L' M=1  

MP2 OUT2 OUT VDD! VDD!  TSMC25P  L='L' W='WPF' AD='WPF*2.5*L' AS='WPF*2.5*L' 

PD='2*WPF+5*L' PS='2*WPF+5*L' M=1  

MN2 OUT2 OUT 0 0  TSMC25N  L='L' W='WNF' AD='WNF*2.5*L' AS='WNF*2.5*L' 

PD='2*WNF+5*L' PS='2*WNF+5*L' M=1  

    

.TRAN  1n 40n START=0 SWEEP DATA=data1 

VIN IN 0 PULSE 0 2.5v  1n 1n 1n 20n 40n 

VVDD! VDD! 0 DC=VDD 

 

.MEASURE TRAN PROPAGATION_R 

+ TRIG v(IN)   VAL=VREF TD=1n FALL=1 

+ TARG v(OUT)  VAL=VREF TD=1n RISE=1 

 

.MEASURE TRAN PROPAGATION_F 

+ TRIG v(IN)   VAL=VREF TD=1n RISE=1 

+ TARG v(OUT)  VAL=VREF TD=1n FALL=1 

 

.MEASURE TRAN PROPAGATION_DELAY PARAM='(PROPAGATION_R+PROPAGATION_F)/2' 

.END 

N o t e :
The simulation output will be stored in the < s p i c e f i l e n a m e > . m t 0  file in a tabular format. You 

can use MATLAB’s s u r f  function to draw your plots. 

 

 

 

The figure above shows the variation of low-to-high propagation delay (tPLH) for an inverter as a 

function of Wp and Wn both. We can observe that with the increase of Wp, the tPLH decreases 

as expected because of the increase of ON current.  

We can also observe that as Wn increases, tPLH also increases. The reason behind is that with 

the increase of Wn, the output node capacitance increases and during the low-to-high 

transition, the increased capacitance results in an increased propagation delay. 

So it quite looks like that both the increased ON current and increased capacitance are playing a 

role for the determination of low-to-high propagation delay (tPLH). 

The figure above shows the variation of high-to-low propagation delay (tPHL) for an inverter as a 

function of Wp and Wn both. We can observe that with the increase of Wn, the tPHL decreases 

as expected because of the increase of ON current.  

We can also observe that as Wp increases, tPHL also increases. The reason behind is that with 

the increase of Wp, the output node capacitance increases and during the high-to-low 

transition, the increased capacitance results in an increased propagation delay. 

So it quite looks like that both the increased ON current and increased capacitance are playing a 

role for the determination of low-to-high propagation delay (tPHL). 

The figure above shows the variation of average propagation delay, (tPHL + tPLH)/2 for an inverter 

as a function of Wp and Wn both. We will consider the following four cases. 1 )
 W p a n d W n b o t h a r e s m a l l : In this case we have lower ON current but also the 

capacitances are small. So we expect lower propagation delays.2 )
 W p a n d W n b o t h a r e h i g h : In this case we have higher ON current but also the 

capacitances are higher. So we expect lower propagation delays.3 )
 W p i s h i g h a n d W n i s s m a l l : In this case we will have lower tPLH but higher tPHL. The 

average value will depend on the exact values of Wn and Wp.4 )
 W p i s l o w a n d W n i s h i g h : In this case we will have higher tPLH but lower tPHL. The 

average value will depend on the exact values of Wn and Wp.

P a r t b )
Let us consider a 2-input NAND gate with inputs A and B. A i s t h e i n p u t t o t h e t r a n s i s t o r t h a t i sc l o s e r t o t h e o u t p u t o f t h e g a t e .  

For low-to-high propagation delay, we can have three scenarios as follows. 

1) B = 1 , A m a k e s t r a n s i t i o n f r o m h i g h - t o - l o w : The pull up PMOS device has to charge only 

the output node capacitance. 

2) A = 1 , B m a k e s t r a n s i t i o n f r o m h i g h - t o - l o w :  The pull up PMOS device has to charge both 

the output node capacitance and the internal node capacitance. 

3) B o t h i n p u t s m a k e t r a n s i t i o n f r o m h i g h - t o - l o w : The effective resistance of the two 

parallel ON PMOS devices is halved than that of the cases when only one PMOS device is 

ON. So the corresponding propagation delay is expected to reduce around half-times. 

According to the above discussion, the 2nd case when A = 1 , B m a k e s t r a n s i t i o n f r o m h i g h - t o - l o w  

is expected to incur a higher low-to-high propagation delay. 

 

For high-to-low propagation delay, we can have three scenarios as follows. 

1) B = 1 , A m a k e s t r a n s i t i o n f r o m l o w - t o - h i g h : The pull up PMOS device has to discharge 

only the output node capacitance. 

2) A = 1 , B m a k e s t r a n s i t i o n f r o m l o w - t o - h i g h :  The pull up PMOS device has to discharge 

both the output node capacitance and the internal node capacitance. 

3) B o t h i n p u t s m a k e t r a n s i t i o n f r o m l o w - t o - h i g h : In this case it is necessary to consider the 

state of the internal node in the NMOS network. The worst case happens when the 

internal node charges up to V D D - V T n  that has to be ensured by setting an initial 

condition of the internal node during simulation. Also when initially both the inputs are 

at l o w values, the o u t p u t h i g h v a l u e  is expected to be at higher value than that of the 

cases when only one input is at l o w values, so a higher voltage at the output node has to 

be brought down to l o w . 

According to the above discussion, the 3rd case when b o t h i n p u t s m a k e t r a n s i t i o n f r o m l o w - t o -h i g h  is expected to incur a higher high-to-low propagation delay. 

H S P I C E C o d e
**** NAND2 maximum rising and falling delays ***** 

* HW 3, Problem 1, Part b) Solution 

* ECE 559, Fall 2009, Purdue University 

 

.TEMP    25.0000 

.lib "$CDK_DIR/models/hspice/public/publicModel/tsmc25N" NMOS  

.lib "$CDK_DIR/models/hspice/public/publicModel/tsmc25P" PMOS  

    

.GLOBAL VDD! 

 

.PARAM VDD  = 2.5 

.PARAM VREF = VDD/2 

.PARAM L    = 300n 

.PARAM WP   = 900n 

.PARAM WN   = 450n 

 

* using FO4 

.PARAM WPF  = '8*WP' 

.PARAM WNF  = '4*WN' 

 

.PARAM RT   = 0.1n 

.PARAM FT   = 0.1n 

 

.OPTION POST 

 

MP1 OUT  IN1 VDD! VDD!  TSMC25P  L='L' W='WP'  

+AD='WP*2.5*L' AS='WP*2.5*L' PD='2*WP+5*L' PS='2*WP+5*L'  

+M=1  

MP2 OUT  IN2 VDD! VDD!  TSMC25P  L='L' W='WP' 

+AD='WP*2.5*L' AS='WP*2.5*L' PD='2*WP+5*L' PS='2*WP+5*L' 

+M=1  

MN1 OUT  IN1 OUTX 0   TSMC25N  L='L' W='WN'  

+AD='WN*2.5*L' AS='WN*2.5*L' PD='2*WN+5*L' PS='2*WN+5*L' 

+M=1  

MN2 OUTX IN2 0    0   TSMC25N  L='L' W='WN'  

+AD='WN*2.5*L' AS='WN*2.5*L' PD='2*WN+5*L' PS='2*WN+5*L' 

+M=1  

   

MP3 OUT2 OUT VDD! VDD!  TSMC25P  L='L' W='WPF'  

+AD='WPF*2.5*L' AS='WPF*2.5*L' PD='2*WPF+5*L' PS='2*WPF+5*L' 

+M=1  

MN3 OUT2 OUT 0 0  TSMC25N  L='L' W='WNF'  

+AD='WNF*2.5*L' AS='WNF*2.5*L' PD='2*WNF+5*L' PS='2*WNF+5*L'  

+M=1  

 

VVDD! VDD! 0 DC=VDD 

VIN1  IN1  0 PAT (VDD 0 1n RT FT 5n b0101110 r=0) 

VIN2  IN2  0 PAT (VDD 0 1n RT FT 5n b0111010 r=0) 

 

* expected values 

VIO    IO  0 PAT (VDD 0 1n RT FT 5n b1010101 r=0)   

 

* set initial condition for 00->11 transition, VDD-Vtn 

.ic OUTX = 1.9882 

 

.tran 1n 35n uic 

.print tran v(IN1) v(IN2) v(IO) v(OUT) v(OUTX) 

 

* measuring the rising delays 

.measure tran propagation_r_11_01 

+ TRIG v(IO)   VAL=VREF TD=1n RISE=1 

+ TARG v(OUT)  VAL=VREF TD=1n RISE=1 

 

.measure tran propagation_r_11_10 

+ TRIG v(IO)   VAL=VREF TD=1n RISE=2 

+ TARG v(OUT)  VAL=VREF TD=1n RISE=2 

 

.measure tran propagation_r_11_00 

+ TRIG v(IO)   VAL=VREF TD=1n RISE=3 

+ TARG v(OUT)  VAL=VREF TD=1n RISE=3 

 

* measuring the falling delays 

.measure tran propagation_f_00_11 

+ TRIG v(IO)   VAL=VREF TD=1n FALL=1 

+ TARG v(OUT)  VAL=VREF TD=1n FALL=1 

 

.measure tran propagation_f_01_11 

+ TRIG v(IO)   VAL=VREF TD=1n FALL=2 

+ TARG v(OUT)  VAL=VREF TD=1n FALL=2 

 

.measure tran propagation_f_10_11 

+ TRIG v(IO)   VAL=VREF TD=1n FALL=3 

+ TARG v(OUT)  VAL=VREF TD=1n FALL=3 

 

.END 

 R e s u l t sR i s i n g D e l a y s
   propagation_r_11_01 (A = 1�0, B = 1) : 2.1521E-10 

   propagation_r_11_10 (A = 1, B = 1�0) : 2.4040E-10 
( m a x i m u m )

 

   propagation_r_11_00 (A = B = 1�0) : 1.1053E-10 

 F a l l i n g D e l a y s
   propagation_f_00_11 (A = B = 0�1) : 2.4176E-10  

( m a x i m u m )
 

   propagation_f_01_11 (A = 0�1, B = 1) : 2.0978E-10 

   propagation_f_10_11 (A = 1, B = 0�1) : 2.2598E-10 

W a v e f o r m

 P a r t c )
In this case we have a 2-input NAND gate and correspondingly we have f o u r  parameters 

(corresponding to f o u r transistors) in the design space. We can use the same widths for the two 

PMOS transistors as the two are connected in parallel. However, since the two NMOS 

transistors in the pull-down network are connected in series, we cannot quite choose the same 

widths for the two NMOS transistors. So we end up in choosing three parameters in design 

space. 

From the part b) answer we know the worst case input vector transitions for the rising and 

falling delays. So we can consider only the worst case input vectors during simulation. We need 

to choose the m a x i m u m  of the rising and falling propagation delays for all the simulated points. 

At last it needs to get the m i n i m u m  value of the maximum propagation delays that will 

correspond to some configuration of the transistor widths in the NAND gate. 

The cadence technology library that we are using restricts us to keep a 75 nm difference 

between two consecutive widths. So we need to choose the set of values in the .DATA 

statement (in the HSPICE netlist) accordingly. 

 H S P I C E C o d e
* HW 3, Problem 1, part c) Solution 

* ECE 559, Fall 2009, Purdue University 

 

.GLOBAL VDD! 

 

.lib "$CDK_DIR/models/hspice/public/publicModel/tsmc25N" NMOS  

.lib "$CDK_DIR/models/hspice/public/publicModel/tsmc25P" PMOS 

 

.PARAM VDD  = 2.5v 

.PARAM VREF = 1.25v 

.PARAM L    = 300n 

 

.PARAM WP   = 450n 

.PARAM WN1  = 450n 

.PARAM WN2  = 450n 

 

* fanout of 4 is chosen 

.PARAM WPF  = '8*WP' 

.PARAM WNF  = '4*WN1' 

 

.PARAM RT   = 1n 

.PARAM FT   = 1n 

 

.TEMP    25.0000 

 

.OPTION  POST    

 

.DATA data1 

 WN1 WN2 WP 

450n 450n 450n 

450n 450n 525n 

450n 450n 600n 

450n 450n 675n 

450n 450n 750n 

450n 450n 825n 

450n 450n 900n 

450n 450n 975n 

* put all set of values that you want to simulate 

.ENDDATA 

 

MP1 OUT IN1 VDD! VDD!  TSMC25P  L='L' W='WP'  

+AD='WP*2.5*L' AS='WP*2.5*L' PD='2*WP+5*L' PS='2*WP+5*L'  

+M=1  

 

MP2 OUT IN2 VDD! VDD!  TSMC25P  L='L' W='WP'  

+AD='WP*2.5*L' AS='WP*2.5*L' PD='2*WP+5*L' PS='2*WP+5*L'  

+M=1  

 

MN1 OUT IN1 OUTX 0  TSMC25N  L='L' W='WN1'  

+AD='WN1*2.5*L' AS='WN1*2.5*L' PD='2*WN1+5*L' PS='2*WN1+5*L' 

+M=1  

 

MN2 OUTX IN2 0 0  TSMC25N  L='L' W='WN2'  

+AD='WN2*2.5*L' AS='WN2*2.5*L' PD='2*WN2+5*L' PS='2*WN2+5*L' 

+M=1  

 

MP3 OUT2 OUT VDD! VDD!  TSMC25P  L='L' W='WPF' AD='WPF*2.5*L' AS='WPF*2.5*L' 

PD='2*WPF+5*L' PS='2*WPF+5*L' M=1  

MN3 OUT2 OUT 0 0  TSMC25N  L='L' W='WNF' AD='WNF*2.5*L' AS='WNF*2.5*L' 

PD='2*WNF+5*L' PS='2*WNF+5*L' M=1  

 

* set initial condition for 00->11 transition, VDD-Vtn 

.ic OUTX = 1.9882 

 

.TRAN  1n 30n START=0 uic SWEEP DATA=data1 

 

*.PRINT TRAN V(IN1) V(IN2) V(OUT) V(OUTX) 

 

VIN1  IN1  0 PAT (VDD 0 0n RT FT 10n b011 r=0) 

VIN2  IN2  0 PAT (VDD 0 0n RT FT 10n b010 r=0) 

 

VVDD! VDD! 0 DC=VDD 

 

.MEASURE TRAN PROPAGATION_R 

+ TRIG v(IN2)  VAL=VREF TD=1n FALL=1 

+ TARG v(OUT)  VAL=VREF TD=1n RISE=1 

 

.MEASURE TRAN PROPAGATION_F 

+ TRIG v(IN2)  VAL=VREF TD=1n RISE=1 

+ TARG v(OUT)  VAL=VREF TD=1n FALL=1 

 

.MEASURE TRAN MAX_PROPAGATION_DELAY 

PARAM='MAX(PROPAGATION_R,PROPAGATION_F)' 

 

.END 

 

 R e s u l t sW N 1 = 4 5 0 n m
(N1 is the transistor closer to the output of the gate)W N 2 = 9 7 5 n mW P = 1 0 5 0 n m

tPLH = 4.0920e-010 sec 

tPHL = 4.1110e-010 sec 

tP.max = max(tPLH,  tPHL) = 4.1110e-010 sec 

 

The reason behind having WN2 > WN1 is quite prominent. If we consider the Elmore delay model 

for the pull-down network, we can easily see that the resistance of the bottom transistor comes 

multiple times in the delay expression. Minimizing the resistance (increasing its width) of the 

bottom transistor hence facilitates a lower delay that has been reflected in the SPICE 

simulation. 

The figure above shows the maximum propagation delay as a function of WN1 and WN2 of the 

simulated NAND gate for Wp = 1050 nm. We can observe that when WN1 = WN2 = 450 nm, the 

delay is higher than that of the case of when WN1 = 450nm, WN2 = 975 nm. We should note that 

increasing the width of the transistor N2 decreases its resistance but consequently increases 

the capacitance. So again we can visualize the trade-off and we can choose the optimal 

transistor widths for propagation delay/performance accordingly. 

P r o b l e m 2 :
Draw a layout of a 2-input NAND gate with the optimized widths that you have got 

for the problem 1. You can round the widths of the transistors as allowed by the technology 

library that you have used in lab. Perform DRC, extract layout, do layout vs. schematic check, 

and at last simulate your extracted netlist to verify its functionality. Submit the plots of 

schematic and layout of your design. 

 S c h e m a t i c D i a g r a m

L a y o u t

 



Homework 3 Solution  ECE 559 (Fall 2009), Purdue University 

 

Page 18 of 22 

 

N o t e :
We can do layout of the higher width transistors in a m u l t i - f i n g e r s t r u c t u r e  as is done for 

the fanout inverter. It facilitates us to keep same spacing between VDD and GND in a row-

based layout structure. P o s t l a y o u t s i m u l a t i o n a n d w a v e f o r m

P r o b l e m 3 :
In the circuit given below, the input voltage, Vin varies as shown. Determine the 

energy dissipation when 1) RC >> T and 2) T >> RC, respectively. R V o u t ( t ) V D D+V i n i ( t ) C
 _          TV i n

0 0

0

in

DD

DD

V t

V
t t T

T
V t T

= ≤

= ≤ ≤

= ≥

 

To determine the total energy dissipation we have to consider the rise time of the input signal 

(0 ≤ t ≤ T) and the ON-period of the input signal separately as we will have different expressions 

for Vout(t) and i(t) for the two cases. 

 

Using KCL 0out out indV V V
C

dt R

−+ =

Taking Laplace transform 

( ),

,

,

( ) ( ) ( ) 0

(1 ) ( ) ( )

( )
( )

(1 )

out out init out in

out in out init

in out init
out

RC SV s V V s V s

SRC V s V s RCV

V s RCV
V s

SRC

− + − =

⇒ + = +
+

⇒ =
+

,

2

2

0 , 0

( ) 1
( )

(1 ) (1 )

1 1 1 1 1 1
1 1

out init

in DD
out

DD DD

For t T V

V s V
V s

SRC T S SRC

V V
RC

T S S T S S
S S

RC RC

≤ ≤ =

= =
+ +

    
    
    = − = − −

       + +              
Taking inverse Laplace transform, 

( ) 1 0
t

DD RC
out

V
V t t RC e for t T

T

−  
= − − ≤ ≤   

  

( )
( ) 1 0

t
out DD RC

dV t V
i t C C e for t T

dt T

− 
= = − ≤ ≤ 

 

,, ( ) 1

1

( )
(1 )

1
1 1

1 1

T

RC
out init out DD DD

T
DD RC

DD DD

out

T

RC
DD DD

DD

RC
For t T V V T V V e

T

V RC
RC V V e

S T
V s

SRC

RC
V V e

T
V

S
S S

RC RC

−

−

−

 
≥ = = − − 

 

  
+ − −   

  =
+

   − −  
  = − +

    + +    
    

 

We have not considered any term like e-ST for u(t-T), instead we would be using (t-T) in place of 

only ‘t’ to signify the fact. 

Taking inverse Laplace transform, 

( ) 1 1

1

t T T t T

RC RC RC
out DD DD DD

T t T

RC RC
DD DD

RC
V t V e V V e e

T

RC
V V e e for t T

T

− −− − −

−− −

    
= − + − −     

    

 
= − − ≥ 

 

( )
( ) 1

T t T
out DD RC RC

dV t V
i t C C e e for t T

dt T

−− − 
= = − ≥ 

 

( )

( ) ( )

0 0 0

0

0

( ) ( ) ( ) ( ) ( ) ( ) ( )

( ) ( ) ( ) ( ) ( ) ( )

( ) 1 ( ) 1

dissipated supply stored

in out in out

T

in out in out

T

T t T t T

RC RC RC
DD DD

T

E E E

i t v t dt i t v t dt i t v t v t dt

i t v t v t dt i t v t v t dt

RC RC
i t V e dt i t V e e dt

T T

CV

∞ ∞ ∞

∞

∞ −− − −

= −

= − = −

= − + −

   
= − + −   

   

=

∫ ∫ ∫

∫ ∫

∫ ∫
2 22 2( )

0

22 2 2( )

0

2 2

1 1

1 2 1
2 2

2 1 1
2

T t T t T
DD RC RC RC

T

Tt t T t T
DD RC RC RC RC

T

T T
DD RC R

R e dt e e dt
T

CV RC RC
R RCe e e e

T

CV RC
R T RC e e

T

∞ −− − −

∞−− − − −

− −

     
 − + −    

       

        = + − + − −      
         

  = − − + −  
   

∫ ∫

2

1
2

T

C RC RC
e

−     + −    

22 2

2

2

2

2 1 1 1
2 2

2 1

1

dissipated

T T T
DD RC RC RC

T T
DD RC RC

T
DD RC

DD

E

CV RC RC
R T RC e e e

T

CV
R T RC e RC RCe

T

CV
R T RC RCe

T

RC RC RC
CV e

T T T

− − −

− −

−

−

        = − − + − + −       
         

    = − − + −   
     

  = − +  
   

 = − + 
 

T

RC
 
 
 

 

2

( )

0.dissipated DD

When T RC T

RC
E CV

T

>> → ∞

 =  
 

≃  

2
2

2 2

( 0)

1
1 1

2

1 1

2 2

dissipated DD

DD DD

When T RC T

RC RC RC T T
E CV

T T T RC RC

RC T
CV CV

T RC

<< →

      − + − +            

  = =  
  

≃

 

We can explain the results intuitively. When the rise time of Vin is low (T << RC), we can assume 

Vin= VDD for the entire transition period and hence the capacitor will charge up to VDD at some 

time provided Vin remains at VDD for sufficient time. So the energy stored in capacitor will be 

(1/2)CVDD
2 and the rest of the energy (1/2)CVDD

2 will be dissipated across the resistor. 

When the rise time of Vin is high (T >> RC), the output signal, Vout can follow the input signal Vin, 

i.e., Vout(t) = Vin(t). Hence, the energy dissipation ~ (Vin(t) - Vout(t)) ~ 0. 


